

Representación gráfica de superficies

Los objetivos de esta práctica son:

- Representar gráficamente superficies en R^3 dadas en forma paramétrica, implícita o en coordenadas cilíndricas o esféricas.
- Representar gráficamente superficies de revolución en R^3 .

1. Superficies en R^3


Superficies en forma explícita

Así como $y=f(x)$ es la forma explícita de describir una curva plana, una superficie en R^3 en forma explícita es la gráfica de una función real de dos variables, es decir, $z=f(x,y)$. Lo mismo que con las curvas, en general se imponen condiciones (por ejemplo, de continuidad) a la función f , que ahora no vamos a tratar. Veamos un ejemplo:

```
> restart:f:=(x,y)->x^2+y^2;
      f:=(x,y)→x2+y2
```

La gráfica de esta función es un subconjunto de R^3 . La representamos mediante la orden **plot3d**:

```
> plot3d(f(x,y),x=-2..2,y=-2..2);
```




Lo mismo que **plot**, esta orden admite muchas opciones. Pero con las gráficas en R^3 tenemos otra manera de modificarlas muy interesante: con el ratón, situamos el cursor sobre la gráfica, picamos y, sin soltarlo, lo movemos en cualquier dirección. También, una vez que hemos picado sobre la gráfica, podemos modificar su aspecto con los botones del menú.

Superficies en forma paramétrica

Son superficies cuyos puntos (x,y,z) vienen dados por $x=f(t,u)$, $y=g(t,u)$, $z=h(t,u)$. Se representan de forma parecida a las curvas planas, con corchetes puestos de distinta manera.

En el ejemplo siguiente, $f(t,u)=t$, $g(t,u)=u-t$, $h(t,u)=t^2+3u^2+8\sin\left(\frac{t^2+u^2}{4}\right)$.

```
> plot3d([t,u-t,t^2+3*u^2+8*sin((t^2+u^2)/4)],t=-4..10,u=-6..6);
```




Superficies en forma implícita

Son superficies cuyos puntos (x,y,z) son soluciones de una ecuación de la forma $f(x,y,z)=0$. Como en el caso de las curvas, hay que imponer ciertas condiciones a la función f para que se trate de lo que entendemos por una superficie. Pero no entraremos aquí en detalles. En Maple se representan mediante la orden **implicitplot3d**. Esta orden pertenece al paquete **plots**.

Por ejemplo, representemos el elipsoide de ecuación $2x^2+3y^2+4z^2=8$.

```
> restart:with(plots):
> implicitplot3d(2*x^2+3*y^2+4*z^2=8,
  x=-2..2,y=-1.7..1.7,z=-1.5..1.5);
```




Si en la orden anterior variamos el rango de las variables x,y,z , la superficie es la misma (o un trozo de ella) pero la resolución con que se dibuja puede variar un poco.

Superficies en coordenadas cilíndricas

De manera análoga al plano, en \mathbb{R}^3 cada punto (x, y, z) se puede representar de la siguiente manera: $x = \rho \cos(\theta)$, $y = \rho \sin(\theta)$, $z = z$. Aquí, $\rho = \sqrt{x^2 + y^2}$ es la distancia del punto al eje z . Estas son las coordenadas cilíndricas. A continuación representamos la superficie $z = \rho^2$, con ρ en el intervalo $[0, 3]$ y θ en el intervalo $[-\frac{3\pi}{4}, \frac{3\pi}{4}]$. Es un trozo de paraboloide.

```
> restart:with(plots):
> implicitplot3d(rho^2=z, rho=0..3, theta=-3*Pi/4..3*Pi/4, z=0..9,
  coords=cylindrical, grid=[15, 15, 15]);
```



Superficies en coordenadas esféricas

Un punto (x, y, z) de \mathbb{R}^3 se puede escribir $x = \rho \cos(\phi) \sin(\theta)$, $y = \rho \sin(\phi) \sin(\theta)$, $z = \rho \cos(\theta)$. Son las coordenadas esféricas. Aquí, ρ es la distancia al origen y se puede encontrar un θ en el intervalo $[0, \pi]$ y un ϕ en el intervalo $[-\pi, \pi]$. Cambiando grados por radianes, θ viene a tener el mismo papel que la latitud de un punto en una esfera (solo que se toma latitud 0 en el polo norte y latitud π en el polo sur), y ϕ tiene el papel de la longitud.

```
> restart:with(plots):
> implicitplot3d(rho=sin(phi)*2^(theta/2),
  theta=-2*Pi..2*Pi, phi=0..Pi, rho=0..5, coords=spherical);
```



Superficies de revolución


En torno al eje de ordenadas

```
> restart:with(plots):
```

Consideremos una curva plana $f: A \rightarrow \mathbb{R}$. Por ejemplo,

```
> f:=x->136*x^3/375-78*x^2/25+107*x/15;
> plot(f(x), x=0..5, y=0..5.1, scaling=constrained);
```


$$f: x \mapsto \frac{136}{375}x^3 - \frac{78}{25}x^2 + \frac{107}{15}x$$



Si imaginamos que la figura gira alrededor del eje de ordenadas (vertical), la curva da lugar a una superficie, que se denomina de revolución (en torno al eje y).

Representemos la curva en \mathbb{R}^3 antes de girar:


```
> spacecurve([0, y, f(y)], y=0..5, scaling=constrained,
  axes=normal, color=red, orientation=[15, 60]);
```



Observemos que hay un cambio de notación, ya que el eje vertical es ahora el eje z , la curva está en el plano zy , y el giro va a ser en torno al eje z . Con este cambio de notación, la curva tiene la ecuación $z = f(y)$, $x = 0$. Es decir, los puntos de la curva son de la forma $(0, y, f(y))$. Además, y pertenece al conjunto A ; en nuestro ejemplo, al intervalo $[0, 5]$.


Un punto $(0, r, f(r))$, al girar, recorre los puntos de la forma (x, y, z) , donde $x^2 + y^2 = r^2$, $z = f(r)$. Dicho de otra forma, la ecuación de la superficie es $z = f(\sqrt{x^2 + y^2})$. Además, hay que imponer que $\sqrt{x^2 + y^2}$ pertenezca al conjunto A; en nuestro ejemplo, al intervalo $[0, 5]$

```
> plot3d(f(sqrt(x^2+y^2)), x=-5..5, y=-sqrt(25-x^2)..sqrt(25-x^2), view=[-5..5, -5..5, 0..5.1], scaling=constrained);
```



También se puede representar la superficie en cilíndricas; en este caso, las ecuaciones que describen la superficie se convierten en $x = \rho \cos(\theta)$, $y = \rho \sin(\theta)$, $z = f(\rho)$.

```
> implicitplot3d(z=f(rho), rho=0..5, theta=0..2*Pi, z=0..5.1, coords=cylindrical, grid=[20, 20, 20]);
```




En torno al eje de abscisas

Ahora hagamos girar también la curva alrededor del eje de abscisas.

Representamos de nuevo la curva antes de girar:

```
> spacecurve([0, y, f(y)], y=0..5, scaling=constrained, axes=normal, color=red, orientation=[15, 60]);
```



Recordemos que, después de un cambio de notación, los puntos de la curva son de la forma $(0, y, f(y))$, donde y pertenece al conjunto A (en nuestro ejemplo, el intervalo $[0, 5]$).

Ahora, el giro es alrededor del eje horizontal y .

Al girar, un punto $(0, r, f(r))$ recorre los puntos (x, y, z) , donde $y = r$, $x^2 + z^2 = f(r)^2$. Dicho de otra forma, la superficie tiene la ecuación $x^2 + z^2 = f(y)^2$. Además, hay que imponer que y pertenezca al conjunto A; en nuestro ejemplo, al intervalo $[0, 5]$.

```
> implicitplot3d(x^2+z^2=f(y)^2, x=-5..5, y=0..5, z=-5..5, scaling=constrained);
```

